
  
               Sql injection filtering    

  

 Sql injectionفیلتر کردن عملیات 
 

 یادآوري این نکته ضروري  اما حتی در هنگام بهترین تنظیمات میگذرد ،sql injectionحدود بیش از ده سال است که از عمر در 
  است که شما همیشه باید به برنامه هاي خود اعتبار بیافزایید 

صرفا براي همه کاربرد ندارد و  براي استفاده بهتر باید آن را  متناسب با نیاز لبی که در طی این مقاله ارائه خواهند شد ، البته مطا
  !خود سازگار کنید 

   )vbscriptاستفاده از  (   نشان دهم را به همهبنابراین من تلاش خواهم کردم تا یک راهکار عمومی 
براي یک مدیریت در ذهن نگه دارید و چیزهایی که ج دارید تا ویژگی هاي وب اپلیکیشن خود را بخطار داشته باشید که شما احتیا

  .صحیح بر وب سرور نیاز دارید را پیوسته تغییر دهید 
  

  !مردم فقط کمی با نحوه فیلترینگ آشنا شوند هدف از این نمونه این است که 
  میکنیمت و کدهاي مخربی را که موجب نفوذ هکرها میشود را فیلتر در اینجا ما یک لیست سیاه تهییه خواهیم کرد و دستورا

  ! استفاده کند UN/**/ION را بصورت union   باز هم شاید هکر بتواند از خلاقیت خود استفاده کند و مثلا  دستوراما
   به هر حال ما فقط راه را نشان خواهیم داد و بقیه خلاقیت ها را به شما واگذار میکنیم ، 

گاهی اوقات خبرنگاران و عده ایی از وب مسترها به من و دوستانم ایراد میگیرند که ! ه دنیاي هک و امنیت دنیاي جالبی است البت
  ! دیگري بلد نیستید ط خرابکاري میکنید و جز نفوذ کارشما ها فق

 شود ، و قاعدتا   عده ایی هم در هر روزه ممکن است آسیب پذیري هایی کشفاما این تصور اشتباه است ، در دنیاي هک و امنیت 
البته سر شار از تجربه و ! ( آیند ، این کارها براي خیلی از متخصصان امنیتی دوست داشتنی و جالب است  صدد امن سازي آن برمی



  !  پیشرفته است  وتقریبا میتوان گفت مانند یک بازي دو طرفه) علم اندوزي 
  

Sql injectionچیست ؟   
  

 استفاده میکند ، هیچگونه ربطی به فایروال یا سایر نرم 80 تزریق کدهاي مخرب است و چون از پورت اینجکشن یک روش
  .  سرور ندارد  درافزارها و سخت افزارهاي امن سازي

  ! استفاده میکند "نشده سازي اعتبار هاي ورودي" براي سو استفاده از آسیب پذیري هاي  موذیانه  روشی از  اینجکشن در واقع 
  ! و با تزریق آن ها به سرور به اطلاعات مهمی دست یابد Sqlک هکر میتواند با اجراي دستورات ی

 news+from+5,password,username,,21+select+union+12=asp.news/com.site.www:  بطور مثال 

  !که در صورت تزریق درست ، نفوذگر به نام کاربري و رمز عبور دست پیدا خواهد کرد 
  !توضیحات بیشتر در زمینه اینجکشن این مقاله را به حاشیه میراند  

  !توضیحاتی کوتاه در مورد اسکریپت 
  

جمع آوري میکنیم و سپس  مخرب استفاده میکنند را در این اسکریپت ابتدا ما یک لیست از کاراکترهایی که هکرها در تزریق کدهاي
را غیر مجاز بشناسد وهکر  url  مجاز ، برنامه urlبرنامه را طوري طراحی میکنیم که در صورت استفاده از این کاراکتر ها بعنوان یک 

  ! هدایت کند error pageرا به صفحه خطلا یا همان  
   )d:حداقل این طوري یکم حرصتون روي هکرا خالی میشه ! ( خطا فحش هم بنویسید اگر مایل باشید میتوانید در صفحه : توضیح 

  

SqlCheckInclude.asp 
  

قرار دهید سپس تغییراتی که متناسب با  Aspاین کد را در یک فایل ! این سورس کد صفحه ایی ست که ورودي ها را تست میکند 
  . نیاز خود میبنید را در سورس کد اعمال کنید 

!                                         فقط لیست سیاه و صفحه خطا را ویرایش کنید و با بقیه کد ها کاري نداشته باشید البته  
 
<%  
'  SqlCheckInclude.asp 
' 
'  Author: d3c0der 
' 
'  This is the include file to use with your asp pages to  
'  validate input for SQL injection. 
 
 
Dim BlackList, ErrorPage, s 
 
' 
'  Below is a black list that will block certain SQL commands and  
'  sequences used in SQL injection will help with input sanitization 
' 
'  However this is may not suffice, because: 
'  1) These might not cover all the cases (like encoded characters) 
'  2) This may disallow legitimate input 
' 
'  Creating a raw sql query strings by concatenating user input is  
'  unsafe programming practice. It is advised that you use parameterized 



'  SQL instead. Check http://support.microsoft.com/kb/q164485/ for information 
'  on how to do this using ADO from ASP. 
' 
'  Moreover, you need to also implement a white list for your parameters. 
'  For example, if you are expecting input for a zipcode you should create 
'  a validation rule that will only allow 5 characters in [0-9]. 
' 
 
BlackList = Array("--", ";", "/*", "*/", "@@", "+",_ 
                  "char", "nchar", "varchar", "nvarchar",_ 
                  "alter", "begin", "cast", "create", "cursor",_ 
                  "declare", "delete", "drop", "end", "exec",_ 
                  "execute", "fetch", "insert", "kill", "open",_ 
                  "select", "sys", "sysobjects", "syscolumns",_ 
                  "table", "update", "union") 
 
'  Populate the error page you want to redirect to in case the  
'  check fails. 
 
ErrorPage = "/ErrorPage.asp" 
                
'''''''''''''''''''''''''''''''''''''''''''''''''''                
'  This function does not check for encoded characters 
'  since we do not know the form of encoding your application 
'  uses. Add the appropriate logic to deal with encoded characters 
'  in here  
''''''''''''''''''''''''''''''''''''''''''''''''''' 
Function CheckStringForSQL(str)  
  On Error Resume Next  
   
  Dim lstr  
   
  ' If the string is empty, return true 
  If ( IsEmpty(str) ) Then 
    CheckStringForSQL = false 
    Exit Function 
  ElseIf ( StrComp(str, "") = 0 ) Then 
    CheckStringForSQL = false 
    Exit Function 
  End If 
   
  lstr = LCase(str) 
   
  ' Check if the string contains any patterns in our 
  ' black list 
  For Each s in BlackList 
   
    If ( InStr (lstr, s) <> 0 ) Then 
      CheckStringForSQL = true 
      Exit Function 
    End If 
   
  Next 
   
  CheckStringForSQL = false 
   
End Function  
 
 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
'  Check forms data 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
 
For Each s in Request.Form 
  If ( CheckStringForSQL(Request.Form(s)) ) Then 
   
    ' Redirect to an error page 
    Response.Redirect(ErrorPage) 



   
  End If 
Next 
 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
'  Check query string 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
 
For Each s in Request.QueryString 
  If ( CheckStringForSQL(Request.QueryString(s)) ) Then 
   
    ' Redirect to error page 
    Response.Redirect(ErrorPage) 
 
    End If 
   
Next 
 
 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
'  Check cookies 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
 
For Each s in Request.Cookies 
  If ( CheckStringForSQL(Request.Cookies(s)) ) Then 
   
    ' Redirect to error page 
    Response.Redirect(ErrorPage) 
 
  End If 
   
Next 
 
 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
'  Add additional checks for input that your application 
'  uses. (for example various request headers your app  
'  might use) 
''''''''''''''''''''''''''''''''''''''''''''''''''' 
 
%> 
  

             TestPage.asp                                                        
                                  

  
    شدن اسکریپت در برنامه است includeاین صفحه نشان دهنده نحوه 

 <%  
'  TestPage.asp 
' 
'  Author: d3c0der 
' 
'  This is a file to test the SQLCheckInclude file. The idea here is that you add 
'  the include file to the beginning of every asp page to get SQL injection  
'  input validation 
 
 
%> 
 
<!--#include file="SqlCheckInclude.asp"--> 
<% 
Response.Write("Welcome to the Test Page.") 
Response.Write("If you are seeing this page then SQL validation succeeded.") 
%> 



  

ErrorPage.asp 
 

  ! این صفحه هدایت میشود این همان صفحه ایی است که در صورت مشکوك قلمداد شدن توسط لیست سیاه ، هکر به
  

<%  
'  ErrorPage.asp 
' 
'  Author: d3c0der 
' 
Response.Write("mage maraz dari ke mikhay site hack koni ? :d ") 
 
'  This is the error page that users will be redirected to if the input cannot 
'  be validated 
  
 
%> 
<%Response.Write("ERROR: Invalid Input")%> 

